
The Dynamic Duo:
ODS Layout and the ODS 
Destination for PowerPoint

Jane Eslinger, SAS Institute Inc.

Today’s Lineup

▪ Game Plan

▪ Playing the Game

▪ Final Stretch

Game Plan:  Understanding Default Behavior

▪ Objects accumulate before being sent to 
PowerPoint

▪ Might place titles and footnotes unexpectedly

▪ Layout container size varies

▪ Might span multiple slides

▪ Might fill just a portion of one slide 

Game Plan:  Triggering a New Slide
▪ Fill the entire slide

▪ Use a DATA _NULL_ step

ods layout gridded;

proc print;

run;

ods layout end;
ods layout gridded;

proc sgplot;

run;

ods layout end;

data _null_;

dcl odsout obj();

run;

Game Plan:  Setting Output Object Size

▪ Graph Size

▪ Set size in GOPTIONS statement

▪ Set size in ODS GRAPHICS statement

▪ Change axis labels and legend

▪ Table Size

▪ Choose best procedure: REPORT, MEANS, TABULATE

▪ Use multiple tables in slide

▪ Remove header rows or change labels

Playing the Game:  Graphs in 2x2 Pattern

proc sort data=sashelp.baseball

out=baseball;

by league division;

run;

Playing the Game:  Graphs in 2x2 Pattern

ods powerpoint file='twobytwo1.pptx'

nogtitle nogfootnote;

title 'Graphs in a 2x2 Pattern';

ods graphics on / height=2.25in width=3in; 

ods layout gridded columns=2 rows=2

advance=bygroup; 

ods region;  

Using the ODS LAYOUT GRIDDED ROWS= option 

Playing the Game:  Graphs in 2x2 Pattern

proc sgplot data=baseball;

by league division;

vbar team / response=nruns;

label league='League'

division='Division';

run;

ods layout end;

ods powerpoint close;

Using the ODS LAYOUT GRIDDED ROWS= option 

Playing the Game:  ODS LAYOUT ROWS=

Playing the Game: Graphs in 2x2 Pattern

ods powerpoint file='twobytwo2.pptx'

nogtitle nogfootnote;

title 'Graphs in a 2x2 Pattern';

ods graphics on / height=2.25in width=3in; 

ods layout gridded columns=2 

advance=bygroup; 

ods region height=47%;

Using the ODS REGION HEIGHT= option

Playing the Game: Graphs in 2x2 Pattern

proc sgplot data=baseball;

by league division;

vbar team / response=nruns;

label league='League'

division='Division';

run;

ods layout end;

ods powerpoint close;

Using the ODS REGION HEIGHT= option

Playing the Game:  ODS REGION HEIGHT= 

Playing the Game:  Multiple Containers

Playing the Game:  Multiple Containers
ods layout gridded rows=1 columns=1;

ods region;

proc odstext;

p 'Table Shows Total Runs and Hits for 

Each League';

p 'The Graph Contains One Bubble for Each

Player. The Size of Each Bubble

Represents the Magnitude of the RBIs.';

run;

ods layout end;

Playing the Game:  Multiple Containers

ods layout gridded columns=2 

column_widths=(47% 47%)column_gutter=1pct;

ods region;

proc tabulate data=sashelp.baseball;

class league;

var nruns nhits nrbi;

tables league='', 

sum='Totals'*(nruns nhits)*f=comma12.;

run;

Playing the Game:  Multiple Containers

ods region;

proc sgplot data=sashelp.baseball;

bubble x=nhits y=nruns size=nrbi/

group=league transparency=.3;

run;

ods layout end;

Playing the Game:  Multiple Containers

Final Stretch

▪ Predefined Layouts

▪ Are specified in ODS POWERPOINT LAYOUT= option

▪ Are not the same as an ODS LAYOUT

▪ Do not use on the same slide as ODS LAYOUT

▪ Possible Values

▪ TitleSlide

▪ TitleAndContent

▪ TwoContent



Square Peg, Square Hole—
Getting Tables to Fit on Slides in the 

ODS Destination for PowerPoint

Jane Eslinger, SAS Institute Inc.



Your task:

Create a PowerPoint
presentation containing
results from your analysis.



Expectations

• Be prepared for trial and error

• Understand some tables will NOT fit on one slide





Default PROC TABULATE Output

ods powerpoint file='example.pptx';

title 'Tabulate: Default';

proc tabulate data=sashelp.prdsal3;

class country product;

var actual predict;

table country*product,(actual predict)*

(n sum mean);

run;

ods powerpoint close;







• Remove optional output
• Reduce table size
• Divide and conquer
• Change font size

Strategies



Strategy:  Remove Optional Output

ods powerpoint file='example.pptx';

title 'Tabulate: NOCONTINUED option';

proc tabulate data=sashelp.prdsal3;

class country product;

var actual predict;

table country*product,(actual predict)*

(n sum mean) / nocontinued;

run;

ods powerpoint close;







Strategy:  Reduce Table Size

ods powerpoint file='example.pptx';

title 'Tabulate: BOX=';

proc tabulate data=sashelp.prdsal3;

class country product;

var actual predict;

table country=''*product='',

(actual predict)*(n sum mean) /

nocontinued box='Country*Product';

run;

ods powerpoint close;







Strategy:  Divide and Conquer

ods powerpoint file='example.pptx';

title 'Tabulate: Page Dimension';

proc tabulate data=sashelp.prdsal3;

class country product;

var actual predict;

table country,product,

(actual predict)*(n sum mean);

run;

ods powerpoint close;







Strategy:  Change Font Size
ods powerpoint file='example.pptx';

title 'Tabulate: Change Font Size';

proc tabulate data=sashelp.prdsal3 style=[fontsize=16pt];

class country product / style=[fontsize=16pt];

classlev country product / style=[fontsize=16pt];

var actual predict / style=[fontsize=16pt];

table country=''*product='',(actual predict)*(n sum mean) /

nocontinued box=[label='Country*Product' 

s=[fontsize=16pt]];

keyword n sum mean / style=[fontsize=16pt];

run;

ods powerpoint close;







Summary

• Remove optional output
• Reduce table size
• Divide and conquer
• Change font size

Jane.Eslinger@sas.com



QUESTIONS?

YES!  I made this presentation with SAS.

I made the peg image with PROC SGPLOT.


